
Chapter 11 Parametric Equations, Polar Curves, and Conic Sections

■ 11.1 Parametric Equations

Students should read Sections 11.1-11.2 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Parametric equations are useful for describing curves not modeled by functions or motions parametrized by quantities such as time. The standard form for a set of parametric equations is

$$\begin{aligned}x(t) &= f(t) \\y(t) &= g(t)\end{aligned}$$

where t is called the *parameter*.

■ 11.1.1. Plotting Parametric Equations


The *Mathematica* command for plotting a curve defined by parametric equations $x = f(t)$ and $y = g(t)$ for $a \leq t \leq b$ is **ParametricPlot**[[$f(t),g(t)$],[t,a,b]].

Here are some examples:

Example 11.1. Plot the curve described by the parametric equations $x = \cos t$ and $y = \sin t$ for $0 \leq t \leq 2\pi$.


Solution:

```
In[276]:= ParametricPlot[{Cos[t], Sin[t]}, {t, 0, 2 π}, ImageSize → 200]
```



NOTE: Recall that the above parametric equations represent the unit circle. However, *Mathematica* may produce a graph that, depending on its default settings, looks visually like an ellipse due to different scalings of the x - and y -axes. In that case, the plot option **AspectRatio** can be used to specify the ratio of the height to the width for a plot. For example, to stretch the plot above so that the circle becomes elliptical where the height is twice as long as the width, we can set **AspectRatio** equal to 2.


In[277]:= `ParametricPlot[{Cos[t], Sin[t]}, {t, 0, 2 π}, AspectRatio → 2, ImageSize → 200]`



Example 11.2. Plot the curve described by the parametric equations $x = t^2 - 4$ and $y = t/2$ for $-2 \leq t \leq 3$.

Solution: Here is a plot of the curve:


In[278]:= `ParametricPlot[{t^2 - 4, t / 2}, {t, -2, 3}]`



Example 11.3. Plot the curve (*prolate cycloid*) described by the parametric equations $x = 2\theta - 4 \sin \theta$ and $y = 2 - 4 \cos \theta$ for $0 \leq \theta \leq 2\pi$ and determine its y -intercepts.

Solution: We first plot the curve:

```
In[279]:= Clear[f, g,  $\theta$ ]
          f[ $\theta_$ ] = 2  $\theta$  - 4 Sin[ $\theta$ ];
          g[ $\theta_$ ] = 2 - 4 Cos[ $\theta$ ];
          ParametricPlot[{f[ $\theta$ ], g[ $\theta$ ]}, { $\theta$ , -4 Pi, 4 Pi},
          PlotLabel -> "Prolate cycloid"]
```



To find the y-intercepts (there appears to be two based on the graph), we solve $f(\theta) = 0$ for θ :

```
In[283]:= Solve[f[ $\theta$ ] == 0,  $\theta$ ]
```

Solve::nsmet : This system cannot be solved with the methods available to Solve. >>

```
Out[283]= Solve[2  $\theta$  - 4 Sin[ $\theta$ ] == 0,  $\theta$ ]
```

Observe that the **Solve** command here fails to give us an answer since the equation is non-algebraic. However, note that the negative y-intercept is trivially located at $y = -2$ corresponding to $\theta = 0$. The other (positive) y-intercept must be solved numerically using the **FindRoot** command (even the **NSolve** command fails in this case), where we provide a nearby location ($\theta = \pi/2$) for our desired solution.

```
In[284]:= root = FindRoot[f[ $\theta$ ] == 0, { $\theta$ , Pi / 2}]
```

```
Out[284]= { $\theta$  -> 1.89549}
```

```
In[285]:= g[root[[1, 2]]]
```

```
Out[285]= 3.27609
```

Thus, the second y-intercept is located approximately at $y \approx f(1.89549) = 3.27609$.

NOTE: Observe that we used of the option **PlotLabel** to print the label "Prolate cycloid" in the plot above. In general, the inside **PlotLabel** \rightarrow "text" inside a plot command prints the title **text** for the given plot.

■ 11.1.2. Parametric Derivatives

Recall that for a curve described by parametric equations $x = f(t)$ and $y = g(t)$, its derivative dy/dx can be expressed as a ratio between the parametric derivatives dy/dt and dx/dt (application of the Chain Rule):

$$\frac{dy}{dx} = \frac{\frac{dy}{dt}}{\frac{dx}{dt}} = \frac{g'(t)}{f'(t)}$$

where it is assumed that $f'(t) \neq 0$.

Example 11.4. Consider the following parametric equations (*folium of Descartes*):

$$x = \frac{4t}{1+t^3} \quad \text{and} \quad y = \frac{4t^2}{1+t^3}$$


- Plot the curve described by the parametric equations above (select an appropriate interval for t that captures all the salient features of the graph).
- Find all points of horizontal tangency on the curve.

c) Find the derivative at the tip of the folium.


Solution:

a) Here is a plot of the folium of Descartes on the interval $[0, 20]$:

```
In[286]:= Clear[f, g, t]
          f[t_] = 3 t / (1 + t^3);
          g[t_] = 3 t^2 / (1 + t^3);
          ParametricPlot[{f[t], g[t]}, {t, 0, 20}, PlotRange -> All, AspectRatio -> 1,
          PlotLabel -> "Folium of Descartes", ImageSize -> 200]
```



NOTE: The plot above does not reveal the full graph of the folium. A more complete graph is shown in the following plot. The dashed line indicates an asymptote. Can you generate a *Mathematica* plot of it? Can you find an equation of the asymptote (see Exercise 7)? Hint: Beware of the discontinuity at $t = -1$.



b) In order to find points of horizontal tangency, that is, points where the slope of the tangent line is equal to zero, it suffices to solve $\frac{dy}{dx} = 0$, or equivalently, $\frac{dy}{dt} = 0$ (assuming $\frac{dx}{dt} \neq 0$). Hence, we evaluate

```
In[290]:= Solve[D[g[t], t] == 0, t]
```

```
Out[290]= {{t -> 0}, {t -> -(-2)^(1/3)}, {t -> 2^(1/3)}, {t -> (-1)^(2/3) 2^(1/3)}}
```

Since $\frac{dx}{dt}$ does not vanish at $t = 0$ and $t = 2^{1/3}$ (we ignore the imaginary solutions), we conclude that there are two points corresponding to these values at which the tangent lines are horizontal, namely at $(0, 0)$ and $(2^{1/3}, 2^{2/3})$:

```
In[291]:= {f[t], g[t]} /. t -> 0
          {f[t], g[t]} /. t -> 2^{1/3}
```

```
Out[291]= {0, 0}
```

```
Out[292]= {2^{1/3}, 2^{2/3}}
```

c) To locate the tip of the folium, we take advantage of the folium's symmetry to argue that the slope of the tangent at the tip must equal -1 (parallel to the asymptote of the folium). Thus, we solve $\frac{dy}{dx} = -1$, or equivalently, $g'(t) = -f'(t)$, for t :

```
In[293]:= sol = Solve[g'[t] == -f'[t], t]
```

```
Out[293]= {{t -> -1}, {t -> 1}}
```

```
In[294]:= f[sol[[1, 1, 2]]]
          g[sol[[1, 1, 2]]]
```

```
Power::infy : Infinite expression  $\frac{1}{0}$  encountered. >>
```

```
Out[294]= ComplexInfinity
```

```
Power::infy : Infinite expression  $\frac{1}{0}$  encountered. >>
```

```
Out[295]= ComplexInfinity
```

Thus, the tip is located at $(3/2, 3/2)$.

■ 11.1.3. Arclength and Speed

The arc length of a curve described by parametric equations $x = f(t)$ and $y = g(t)$, $a \leq t \leq b$, is given by

$$s = \int_a^b \sqrt{\left(\frac{dx}{dt}\right)^2 + \left(\frac{dy}{dt}\right)^2} dt$$

Suppose $(x(t), y(t))$ now represent the position of a particle moving along a path at time t . The distance traveled by the particle over the interval $[t_0, t]$ is given by

$$s(t) = \int_{t_0}^t \sqrt{\left(\frac{dx}{du}\right)^2 + \left(\frac{dy}{du}\right)^2} du$$

and


$$\frac{ds}{dt} = \sqrt{\left(\frac{dx}{dt}\right)^2 + \left(\frac{dy}{dt}\right)^2}$$

represents its speed (length of the velocity vector $(x'(t), y'(t))$).

Example 11.5. Find the arc length of the curve $x = e^{-t} \cos t$, $y = e^{-t} \sin t$ for $0 \leq t \leq \pi/2$.

Solution: Here is a plot of the curve:

```
In[296]:= Clear[x, y, t]
          x[t_] = E^(-t) * Cos[t];
          y[t_] = E^(-t) * Sin[t];
          ParametricPlot[{x[t], y[t]}, {t, 0, Pi/2}, ImageSize -> 250]
```



To find its arc length, we compute

```
In[300]:= ∫₀^{π/2} √((D[x[t], t])² + (D[y[t], t])²) dt
```

```
Out[300]= √2 (1 - e^{-π/2})
```

```
In[301]:= N[%]
```

```
Out[301]= 1.12023
```

This answer makes sense based on the graph.

Example 11.5. A particle moves along a path described by $x = t^2 - 4$, $y = t/2$.

- Find the distance traveled by the particle over the interval $[-3, 3]$.
- What is the particle's minimum speed over the same interval? When does this occur?

Solution:

a) The distance traveled is given by

```
In[302]:= Clear[x, y, t]
          x[t_] = t^2 - 4;
          y[t_] = t/2;
          ∫_{-3}^3 √((D[x[t], t])² + (D[y[t], t])²) dt
```

```
Out[305]= 1/8 (12 √145 + ArcSinh[12])
```


```
In[306]:= N[%]
```

```
Out[306]= 18.4599
```

a) We make a plot of the particle's speed:

```
In[307]:= speed =  $\sqrt{((D[x[t], t])^2 + (D[y[t], t])^2)}$ 
Plot[speed, {t, -3, 3}, ImageSize -> 250]
```

```
Out[307]=  $\sqrt{\frac{1}{4} + 4 t^2}$ 
```



This shows that the minimum speed is $1/2$ and occurs at $t = 0$. Can you verify this using calculus techniques?

■ Exercises

In Exercises 1 through 3, sketch the curve represented by the given parametric equations. Be sure to select an appropriate interval for the parameter that captures all the salient features of the curve.

1. $x = t^3, y = t^2/2$ 2. $x = 2(\theta - \sin \theta), y = 1 - \cos \theta$ 3. $x = 3 \cos^3 \theta, y = 3 \sin^3 \theta$

4. Find all points of horizontal and vertical tangency to the curve $x = \cos \theta + \theta \sin \theta, y = \sin \theta - \theta \cos \theta, 0 \leq \theta \leq 2\pi$.

5. Consider parametric equations given by $x = 3 \cos(t/3) - \cos t$ and $y = 3 \sin(t/3) - \sin t$.

a) Graph the curve represented by the parametric equations above.

b) Find the slope of the line tangent to the curve at the point where $t = \pi/4$.

c) Find the arc length of the curve from $t = 0$ to $t = 3\pi/2$.

6. Consider a particle moving along a curve described by $x = t - \cos t$ and $y = t - \sin t$ with respect to time t .

a) Approximate the distance traveled by the particle over the interval $[0, 2\pi]$. Hint: Use the **NIntegrate** command.

b) Find the minimum and maximum speeds of the particle over the same interval. At what times do they occur?

7. *Cornu's spiral* (also known as Euler's spiral) is a curve defined by the following Fresnel integrals:

$$x(t) = \int_0^t \cos(u^2) du, \quad y(t) = \int_0^t \sin(u^2) du$$

a) Plot Cornu's spiral for $-10 \leq t \leq 10$.

b) Compute the length of Cornu's spiral over the same interval. Then find a formula for the length of Cornu's spiral over the interval $a \leq t \leq b$.

c) Determine the coordinates of the center of the two "eyes" that form Cornu's spiral.

8. Find the asymptote corresponding to the folium of Descartes (see Example 11.4) and plot the asymptote together with the folium of Descartes.

■ 11.2 Polar Coordinates and Curves

Students should read Sections 11.3-11.4 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in

this section.

Polar coordinates, expressed as (r, θ) , represent the location of a point on the Cartesian plane in terms of distance r from the origin and angle θ with respect to the positive x -axis.

11.2.1. Conversion Formulas

Conversion between Cartesian (rectangular) coordinates (x, y) and polar coordinates (r, θ) can be achieved by the following formulas:

$$\begin{aligned} r^2 &= x^2 + y^2 & x &= r \cos \theta \\ \theta &= \tan^{-1} y/x & y &= r \sin \theta \end{aligned}$$

Example 11.6. Perform the following conversions:

- Convert the rectangular coordinates $(3, 4)$ into polar coordinates.
- Convert the polar coordinates $(7, \pi/3)$ into Cartesian coordinates.

Solution:

a) Using the first set of conversion formulas above, we find that

```
In[309]:= r = Sqrt[3^2 + 4^2]
          theta = N[ArcTan[4 / 3]]
```

```
Out[309]= 5
```

```
Out[310]= 0.927295
```

b) This time, we use the second set of conversion formulas:

```
In[311]:= x = 7 * Cos[Pi / 3]
          y = 7 * Sin[Pi / 3]
```

```
Out[311]= 7/2
```

```
Out[312]= (7*Sqrt[3])/2
```


11.2.2. Polar Curves

The *Mathematica* command for plotting a curve described by a polar equation in the form $r = f(\theta)$ for $\alpha \leq \theta \leq \beta$ is **PolarPlot** $[f(\theta), \{\theta, \alpha, \beta\}]$.

Example 11.7. Plot the graph of the limaçon $r = 3 - 4 \cos \theta$.

Solution: Here is a plot of the limaçon:



```
In[313]:= PolarPlot[3 - 4 Cos[θ], {θ, -4 Pi, 4 Pi},
  AspectRatio -> Automatic, ImageSize -> 200]
```



Example 11.8. Plot the graph of the six-leaf rose $r = 2 \cos(3\theta/2)$.

Solution: Here is a plot of the six-leaf rose:

```
In[314]:= PolarPlot[2 Cos[3 θ / 2], {θ, -4 Pi, 4 Pi},
  PlotLabel -> "A Six-Leaf Rose", AspectRatio -> Automatic, ImageSize -> 200]
```



Can you can modify the function to generate a 12-leaf rose?

■ 11.2.2. Calculus of Polar Curves

Recall that the derivative of a polar equation in the form $r = f(\theta)$ for $\alpha \leq \theta \leq \beta$ is given by

$$\frac{dy}{dx} = \frac{f'(\theta) \sin \theta + f(\theta) \cos \theta}{f'(\theta) \cos \theta - f(\theta) \sin \theta}$$


Moreover, the area A of the region bounded by a polar equation in the form $r = f(\theta)$ between $\alpha \leq \theta \leq \beta$ is given by

$$A = \frac{1}{2} \int_{\alpha}^{\beta} f^2(\theta) d\theta$$

Example 11.9. Locate all horizontal and vertical tangents of the limaçon $r = 2 - \sin \theta$.

Solution: We first plot the limaçon to anticipate our solution points:

```
In[315]:= PolarPlot[1 + Cos[θ], {θ, -2 Pi, 2 Pi}, ImageSize → 200]
```



From the plot above, we should expect to find two horizontal tangents and three vertical tangents.

Next, we compute the derivative of the limaçon:

```
In[316]:= Clear[f, θ]
```

```
f[θ_] = 1 + Cos[θ]
```

```
Out[317]= 1 + Cos[θ]
```

```
In[318]:= dydx = Simplify[(f'[θ] * Sin[θ] + f[θ] * Cos[θ]) / (f'[θ] * Cos[θ] - f[θ] * Sin[θ])]
```

```
Out[318]= -Cos[3θ/2] Csc[θ/2] / (1 + 2 Cos[θ])
```

To obtain horizontal tangents, we solve $dy/dx = 0$ for θ .

```
In[319]:= dydx == 0
```

```
Solve[dydx == 0, θ]
```

```
Out[319]= -Cos[3θ/2] Csc[θ/2] / (1 + 2 Cos[θ]) == 0
```

Solve::ifun : Inverse functions are being used by Solve, so

some solutions may not be found; use Reduce for complete solution information. >>

```
Out[320]= {{θ → -5π/3}, {θ → -π}, {θ → -π/3}, {θ → π/3}, {θ → π}, {θ → 5π/3}}
```

Therefore, our two horizontal tangents are located at $\theta = \pm\pi/3$.

As for vertical tangents, we solve for where the reciprocal of the derivative is zero, that is, $1/(dy/dx) = 0$ for θ .

In[321]:= **1 / dydx**

Solve[1 / dydx == 0, θ]

Out[321]= $-(1 + 2 \cos[\theta]) \sec\left[\frac{3\theta}{2}\right] \sin\left[\frac{\theta}{2}\right]$

Solve::ifun : Inverse functions are being used by Solve, so

some solutions may not be found; use Reduce for complete solution information. >>

Out[322]= $\left\{\left\{\theta \rightarrow 0\right\}, \left\{\theta \rightarrow -\frac{4\pi}{3}\right\}, \left\{\theta \rightarrow -\frac{2\pi}{3}\right\}, \left\{\theta \rightarrow \frac{2\pi}{3}\right\}, \left\{\theta \rightarrow \frac{4\pi}{3}\right\}\right\}$


Since the solution above only gives us the principal solution $\theta = 0$ of $-\tan(3\theta/2) = 0$, we need to additionally solve $3\theta/2 = \pm\pi$ for θ , which yields our two other solutions, $\theta = \pm 2\pi/3$.

NOTE: What is the derivative at $\theta = \pi$?

Example 11.10. Find the area of the region contained inside the circle $r = 3 \sin \theta$ and outside the convex limaçon $r = 2 - \sin \theta$.

Solution: We first plot the two polar curves on the same set of axes.

In[323]:= **PolarPlot[{3 Sin[θ], 2 - Sin[θ]}, { θ , 0, 2 π },
AspectRatio -> Automatic, ImageSize -> 200]**



Next, we find their points of intersection by equating them and solving for θ :

In[324]:= **Solve[3 Sin[θ] == 2 - Sin[θ], θ]**

Solve::ifun : Inverse functions are being used by Solve, so

some solutions may not be found; use Reduce for complete solution information. >>

Out[324]= $\left\{\left\{\theta \rightarrow \frac{\pi}{6}\right\}\right\}$

Observe that *Mathematica* gives only the solution $\theta = \pi/6$, which lies in the first quadrant since trigonometric inverse functions are involved. We can see from the above graph that the other point of intersection must be at $\theta = 5\pi/6$ due to symmetry. Thus, the area of the enclosed region is given by the difference in areas enclosed by the circle and limaçon between $\theta = \pi/6$ and

$$\theta = 5\pi/6:$$

```
In[325]:= (1/2) (Integrate[(3 Sin[θ])^2, {θ, Pi/6, 5 Pi/6}] -
           Integrate[(2 - Sin[θ])^2, {θ, Pi/6, 5 Pi/6}])
```

```
Out[325]= 3 √3
```

```
In[326]:= N[%]
```

```
Out[326]= 5.19615
```

NOTE: Using even symmetry of our region, it would have been enough to integrate between $\theta = \pi/6$ and $\theta = \pi/2$ and double the result.

■ Exercises

In Exercises 1 and 2, use *Mathematica* to perform the following conversions.

1. Convert the rectangular coordinates $(-1, \sqrt{3})$ into polar coordinates.
2. Convert the polar coordinates $(5, 3\pi/4)$ into Cartesian coordinates. What if we replace $(5, 3\pi/4)$ with $(-5, 3\pi/4)$?

In Exercises 3 through 6, plot the graph of each of the given polar equations and find an interval for θ over which each graph is traced only once.

3. $r = 3 - 4 \cos \theta$ 4. $r = 2 + \sin \theta$ 5. $r = 3 \cos(3\theta/2)$ 6. $r = 5 \sin 2\theta$

7. Generate the butterfly curve $r = e^{\cos \theta} - 2 \cos(4\theta) + \sin^5(\theta/12)$.

8. Find all horizontal and vertical tangents of the lemniscate $r^2 = \cos(2\theta)$. Plot its graph to confirm your answers.

9. Consider the rose curve $r = \cos(2\theta)$ for $-2\pi \leq \theta \leq 2\pi$.

- a) Plot its graph.
- b) Find the area of one petal of the curve.

In Exercises 10 through 12, graph and find the area of each of the given regions.

10. The common interior of $r = 3 - 2 \sin \theta$ and $r = -3 + 2 \sin \theta$.

11. Inside $r = 2(1 + \cos \theta)$ and outside $r = 2 \cos \theta$.

12. Inner loop of $r = 3 + 4 \sin \theta$.

In Exercises 13 and 14, find the length of the given curve over the specified interval.

13. $r = 1 + \sin \theta, 0 \leq \theta \leq 2\pi$ 14. $r = 6(1 + \cos \theta), 0 \leq \theta \leq 2\pi$

15. Consider the polar equations $r = 4 \sin \theta$ and $r = 2(2 - \sin^2 \theta)$.

- a) Graph the polar equations on the same axes.
- b) Find the points of intersection of the curves.
- c) Find the circumference of each curve.

■ 11.3 Conic Sections

Students should read Section 11.5 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Conic sections refer to the three families of curves (ellipses, hyperbolas, parabolas) generated by intersecting a plane with a cone. Recall the equations for describing each family of curves in standard position:

I. Ellipse

$$\left(\frac{x}{a}\right)^2 + \left(\frac{y}{b}\right)^2 = 1$$

If $b > a > 0$, then the ellipse has two foci located at $(0, \pm c)$, where $c = \sqrt{b^2 - a^2}$.

II. Hyperbola

$$\left(\frac{x}{a}\right)^2 - \left(\frac{y}{b}\right)^2 = 1$$

III. Parabola

$$y = \frac{1}{4c^2} x^2$$

NOTE: These formulas assume that the "center" of the conic section is at the origin. To translate the center to a different point, say (x_0, y_0) , we replace x and y by $x - x_0$ and $y - y_0$, respectively.

The most useful command for plotting conic sections is **ContourPlot[eqn,{x,a,b},{y,c,d}]**, where **eqn** is the equation of the conic.

Example 11.10. Determine the family that each conic section below belongs to and then make a plot of each.

a) $\frac{x^2}{9} + \frac{y^2}{16} = 1$


b) $y = \frac{9}{4} x^2$

c) $\frac{x^2}{4} - \frac{y^2}{9} = 1$

Solution:

a) This conic is an ellipse. To plot it, we evaluate


```
In[327]:= ContourPlot[x^2/9 + y^2/16 == 1, {x, -4, 4},
             {y, -4, 4}, Axes -> True, Frame -> False, ImageSize -> 200]
```



Observe that the length of semi-major and semi-minor axes are 4 and 3, respectively. How would this change if we happen to switch the coefficients 9 and 16?


b) This conic is a parabola. Since the equation here is solved for y , we merely use the **Plot** command:

In[328]:= `Plot[(9/4) x^2, {x, -2, 2}, ImageSize -> 200]`



c) This conic is a hyperbola. Here is its plot:

In[329]:= `ContourPlot[x^2/4 - y^2/9 == 1, {x, -6, 6}, {y, -6, 6}, Axes -> True, Frame -> False, ImageSize -> 200]`




Example 11.11. Find an equation of an ellipse with center at $(-1, 3)$ and having semi-major and semi-minor axes of lengths $\sqrt{5}$ and $1/2$, respectively.

Solution: From the given data, we see that $(x_0, y_0) = (-1, 3)$, $a = \sqrt{5}$, and $b = 1/2$. The equation of our ellipse is therefore:

$$\frac{(x+1)^2}{5} + 4(y-3)^2 = 1$$

To plot it, we evaluate

In[330]:= `ContourPlot[(x+1)^2/5 + 4(y-3)^2 == 1, {x, -4, 2}, {y, 2, 4}, AspectRatio -> 2/3, Axes -> True, Frame -> False, ImageSize -> 200]`



■ Exercises

In Exercises 1 through 4, plot each of the given conic sections. Can you determine the family that each conic section belongs to before plotting? Also, what are the values of the parameters a and b (or c) in each conic section?

$$1. \frac{x^2}{25} + \frac{y^2}{4} = 1 \quad 2. \frac{y^2}{36} - \frac{x^2}{16} = 1 \quad 3. y = 3x^2 \quad 4. \frac{(y-2)^2}{25} - \frac{(x+1)^2}{49} = 1$$

$$5. \text{ Consider the ellipse } \frac{x^2}{9} + 4y^2 = 1.$$

- Make a plot of the ellipse.
- What are the lengths of the semi-major and semi-minor axes?
- Where are the foci located?
- Compute the sum of the distances from the two foci to any point on the ellipse. Do you recognize the answer?

6. Find an equation of an ellipse with center $(1/2, -5)$ and having semi-major and semi-minor axes of lengths $3/4$ and $\sqrt{7}$, respectively.

7. Find all points on the hyperbola $\frac{x^2}{25} - \frac{y^2}{9} = 1$ where its slope equals 1.

8. Consider a polar curve of the form $r = \frac{de}{1+e \cos \theta}$, where d and e are non-negative constants.

- Plot this curve for $d = 3$, and $e = 1/2$. Do you recognize this curve as a conic section? Of which type? Hint: Use the command **PolarPlot**.
- Repeat part a) but this time use $e = 2$ instead. Do you recognize this curve as a conic section? Of which type?
- Repeat part a) but this time use $e = 1$ instead. Do you recognize this curve as a conic section? Of which type?
- Describe how the graph changes as we vary the values d and e . What happens to the graph when $e = 0$? NOTE: The value e is called the *eccentricity* of the conic section.
- Assume $0 < e < 1$. Convert the polar equation $r = \frac{de}{1+e \cos \theta}$ to that in standard form for an ellipse, $\left(\frac{x-x_0}{a}\right)^2 + \left(\frac{y-y_0}{b}\right)^2 = 1$, and determine formulas for its center, semi-major, and semi-minor axes. Verify these formulas for the ellipse in part a).